
#include <Wire.h>
byte bus_address = 0x64;

char computerdata[30];
char *cmd; 
char *data_byte_1;
byte received_from_computer = 0;
byte serial_event = 0;
byte number_of_bytes; 
byte i2c_device[37];

//enable I2C
//holds I2C address. The default address is (0x64) 100

void setup() {
  Serial.begin(9600);
  Wire.begin();
}

//hardware setup
//enable serial port at 9600 baud
//enable I2C port

void serialEvent() {
  received_from_computer = Serial.readBytesUntil(13, computerdata, 20);
  computerdata[received_from_computer] = 0;
  serial_event = 1;
}

void pars_data() {                                                      //once a string is received from the serial monitor it is parsed at each comma
  cmd = strtok(computerdata, ",");                           //let's pars the string at each comma
  data_byte_1 = strtok(NULL, ",");                           //let's pars the string at each comma
}

void i2c_read(byte reg,  byte number_of_bytes_to_read) {              
  byte i; 
  Wire.beginTransmission(bus_address); 
  Wire.write(reg); 
  Wire.endTransmission();

  Wire.requestFrom(bus_address, (byte)number_of_bytes_to_read);  
  for (i = reg; i < (reg + number_of_bytes_to_read); i++) { 
    i2c_device[i] = Wire.read();                                      
  }
  Wire.endTransmission();                                             
}

unsigned int read_two_bytes(byte reg) {                                //this function will take 2 bytes from the array i2c_device[x], join them together and return an unsigned int
  uint16_t two_byte_data;                                               //an unsigned 16-bit integer
  two_byte_data = i2c_device[reg];                                    //load in the first byte(MSB)
  two_byte_data = two_byte_data << 8;                                 //shift the first byte over 8 places
  two_byte_data = two_byte_data | i2c_device[reg += 1];            //move to the next element in the array and load in the second byte
  return two_byte_data;                                                //return the new number
}

unsigned long read_four_bytes(byte reg) {                              //this function will take 4 bytes from the array i2c_device[x], join them together and return an unsigned long
  uint32_t four_byte_data;                                                //an unsigned 32-bit long
  four_byte_data = i2c_device[reg];                                         //load in the first byte(MSB)
  four_byte_data = four_byte_data << 8;                                  //shift the bytes over 8 places
  four_byte_data = four_byte_data | i2c_device[(reg += 1)];             //move to the next element in the array and load in the second byte
  four_byte_data = four_byte_data << 8;                                   //shift the bytes over 8 places
  four_byte_data = four_byte_data | i2c_device[(reg += 1)];               //move to the next element in the array and load in the third byte
  four_byte_data = four_byte_data << 8;                                  //shift the bytes over 8 places
  four_byte_data = four_byte_data | i2c_device[(reg += 1)];             //move to the next element in the array and load in the fourth byte
  return four_byte_data;                                                   //return the new number
}

void four_byte_write(uint8_t reg, int32_t _value) {                   //this function will take a four byte number and send it byte by byte to the device
  Wire.beginTransmission(bus_address);                                    //call the device by its ID number
  Wire.write(reg);                                                        //transmit the register that we will start from
  Wire.write(_value >> 24);                                            //send the MSB
  Wire.write((_value >> 16) & 0xFF);                                 //send the high byte
  Wire.write((_value >> 8) & 0xFF);                                  //send the low byte
  Wire.write(_value & 0xFF);                                            //send the LSB
  Wire.endTransmission();                                              //end the I2C data transmission
}

void two_byte_write(uint8_t reg, uint16_t _value) {                //this function will take a two byte number and send it byte by byte to the device
  Wire.beginTransmission(bus_address);                                   //call the device by its ID number
  Wire.write(reg);                                                                      //transmit the register that we will start from
  Wire.write((_value >> 8) & 0xFF);                                          //send the MSB
  Wire.write(_value & 0xFF);                                                    //send the LSB
  Wire.endTransmission();                                                         //end the I2C data transmission
}

void one_byte_write(uint8_t reg, uint8_t _value) {                //this function will send one byte to the device
  Wire.beginTransmission(bus_address);                                 //call the device by its ID number
  Wire.write(reg);                                                                    //transmit the register that we will start from
  Wire.write(_value);                                                   //send the byte
  Wire.endTransmission();                                              //end the I2C data transmission
}

void loop() {                                                        //main loop

  if (serial_event) {                                               //if the serial_event flag is set action is taken
    serial_event = 0;                                             //reset the flag
    pars_data();                                                    //calling this function will break up the ASCII string command sent to the Arduino
    read_command();                                            //read the command sent and take some kind of action
  }
}

void read_command() {                                       //this function will read what command was sent

  if (strcmp(cmd, "?") == 0) {                              //if a "?" was sent
    device_commands();                                       //call the function "device_commands" this will display what commands will work with this sample code
  }

  if (strcmp(cmd, "i") == 0) {                               //if an "i" was sent
    device_info();                                                  //call the function "device_info" this will display the status of registers 0x00 to 0x13
  }

 if (strcmp(cmd, "k") == 0) {                               //i an "k," was sent
    float probe;                                                    //make a var to hold a floating point number
    unsigned int new_probe;                                //make a var to hold an unsigned int

    probe = atof(data_byte_1);                            //convert the character pointer data_byte_1 from a string to a float and store it in a var called probe
    probe *= 100;                                                 //multiply by 100 to remove the floating point number
    new_probe = probe;                                      //move the float to a int
    two_byte_write(0x08, new_probe);               //write the int to the device  starting at register 0x08
    Serial.println("*OK");                                      //"*OK"
  }

if (strcmp(cmd, "t") == 0) {                                //if an "t," was sent
    float temperature;                                         //make a var to hold a floating point number
    unsigned long new_temperature;                  //make a var to hold an unsigned long

    temperature = atof(data_byte_1);                  //convert the character pointer data_byte_1 from a string to a float and store it in a var called temperature
    temperature *= 100;                                      //multiply by 100 to remove the floating point number
    new_temperature = temperature;                 //move the float to a long
    four_byte_write(0x10, new_temperature);    //write the long to the device  starting at register 0x10
    Serial.println("*OK");                                     //"*OK"
  }

  if (strcmp(cmd, "calval") == 0) {                       //if "calval," was sent
    float cal_val;                                                   //make a var to hold a floating point number
    unsigned long new_cal_val;                            //make a var to hold an unsigned long

    cal_val = atof(data_byte_1);                           //convert the character pointer data_byte_1 from a string to a float and store it in a var called cal_val
    cal_val *= 100;                                                //multiply by 100 to remove the floating point number
    new_cal_val = cal_val;                                     //move the float to a long
    four_byte_write(0x0A, new_cal_val);             //write the long to the device starting at register 0x0A
    Serial.println("*OK");                                      //"*OK"
  }

if (strcmp(cmd, "cal") == 0) {                              //if "cal,[command]" was sent

    if (strcmp(data_byte_1, "clear") == 0) {         //if "cal,clear" was sent
      one_byte_write(0x0E, 1);                             //write a 1 to register 0x0E to clear the calibration
      Serial.println("*OK");                                   //"*OK"
    }

    if (strcmp(data_byte_1, "dry") == 0) {           //if "cal,dry" was sent
      one_byte_write(0x0E, 2);                             //write a 2 to register 0x0E to dry calibrate
      Serial.println("*OK");                                   //"*OK"
    }

    if (strcmp(data_byte_1, "one") == 0) {           //if "cal,one" was sent
      one_byte_write(0x0E, 3);                             //write a 3 to register 0x0E to do a single point calibration
      Serial.println("*OK");                                   //"*OK"
    }

    if (strcmp(data_byte_1, "low") == 0) {           //if "cal,low" was sent
      one_byte_write(0x0E, 4);                             //write a 4 to register 0x0E to do the low point calibration of a two point calibration
      Serial.println("*OK");                                    //"*OK"
    }

    if (strcmp(data_byte_1, "high") == 0) {          //if "cal,high" was sent
      one_byte_write(0x0E, 5);                             //write a 5 to register 0x0E to do the low point calibration of a two point calibration
      Serial.println("*OK");                                   //"*OK"
    }
  }

if (strcmp(cmd, "on") == 0) {                              //if "on" was sent
    one_byte_write(0x05, 8);                               //set the LED to toggle it start after each reading
    one_byte_write(0x06, 1);                               //wake up the device so it starts taking readings
    Serial.println("*OK");                                      //"*OK"
  }

  if (strcmp(cmd, "off") == 0) {                            //if "off" was sent
    one_byte_write(0x05, 0);                               //turn off the LED
    one_byte_write(0x06, 0);                               //place the device in hibernation
    Serial.println("*OK");                                      //"*OK"
  }

 if (strcmp(cmd, "r") == 0) {                                //if an "r" was sent
    unsigned long ec;                                           //make a var to hold an unsigned long
    unsigned long tds;                                          //make a var to hold an unsigned long
    unsigned long pss;                                          //make a var to hold an unsigned long
    i2c_read(0x18, 12);                                         //read some device registers, start at register 0x18 and read 12 bytes
    ec = read_four_bytes(0x18);                           //starting at register 0x18 combine 4 bytes into an unsigned long
    tds = read_four_bytes(0x1C);                         //starting at register 0x1C combine 4 bytes into an unsigned long
    pss = read_four_bytes(0x20);                         //starting at register 0x20 combine 4 bytes into an unsigned long

    Serial.println("");
    Serial.print("EC= ");
    Serial.println((float)ec / 100);                         //just before we display the reading cast the long into a float and divide by 100
    Serial.print("TDS= ");
    Serial.println((float)tds / 100);                        //just before we display the reading cast the long into a float and divide by 100
    Serial.print("Salinity= ");
    Serial.println((float)pss / 100);                        //just before we display the reading cast the long into a float and divide by 100
    Serial.println("");
  }
}

void device_info() {                                             //when the "i" command is sent this function is called
                                                                           //it reads all status registers and prints out the data
  unsigned int probe_type;                                 //make a var to hold an unsigned int
  unsigned long cal_val;                                      //make a var to hold an unsigned long
  unsigned long temp_comp;                              //make a var to hold an unsigned long

  i2c_read(0x00, 20);                                           //read some device registers, start at register 0x00 and read 20 bytes

  Serial.println("");                                              //this helps make the output easier to read
  Serial.println("");                                              //this helps make the output easier to read

  Serial.print("Device Type: ");
  if (i2c_device[0] == 4) {                                     //all the device registers are read into an array. The array number matches the device register number
    Serial.println("EC OEM");                               //register 0x00 is the device type register so, i2c_device[0] hold the byte from that register  
  }
  else Serial.println("unknown");

  Serial.print("Device Version: ");
  Serial.println (i2c_device[1]);                            //array [1] = register 0x01

  Serial.print("Address Lock/Unlock: ");
  if (i2c_device[2] == 1) {                                     //array [2] = register 0x02                              
    Serial.println("Locked");
  }
  if (i2c_device[2] == 0) {
    Serial.println("Unlocked");
  }

  Serial.print("I2C Address: ");
  Serial.println (i2c_device[3]);                            //array [3] = register 0x03 

  Serial.print("Interrupt Control: ");
  if (i2c_device[4] == 0) {
    Serial.println("Disabled");
  }
  if (i2c_device[4] == 1) {
    Serial.println("Pin high");
  }
  if (i2c_device[4] == 2) {
    Serial.println("High on new reading");
  }
  if (i2c_device[4] == 4) {
    Serial.println("Low on new reading");
  }
  if (i2c_device[4] == 8) {
    Serial.println("invert on new reading");
  }

  Serial.print("LED Control: ");
  if (i2c_device[5] == 0) {
    Serial.println("Disabled");
  }
  if (i2c_device[5] == 1) {
    Serial.println("On");
  }
  if (i2c_device[5] == 2) {
    Serial.println("LED on upon new reading");
  }
  if (i2c_device[5] == 4) {
    Serial.println("LED off upon new reading");
  }
  if (i2c_device[5] == 8) {
    Serial.println("LED invert upon new reading");
  }

  Serial.print("Active/Hibernate: ");
  if (i2c_device[6] == 1) {
    Serial.println("System on");
  }
  if (i2c_device[6] == 0) {
    Serial.println("System Hibernate");
  }

  Serial.print("New Reading Available: ");
  if (i2c_device[7] == 1) {
    Serial.println("Yes");
  }
  if (i2c_device[7] == 0) {
    Serial.println("No");
  }

  Serial.print("Probe type: ");
  Serial.print("K=: ");
  probe_type = read_two_bytes(0x08);
  Serial.println((float)probe_type / 100);

  Serial.print("Calibration Value: ");
  cal_val = read_four_bytes(0x0A);
  Serial.println((float)cal_val / 100);

  Serial.print("Calibration Confirm: ");
  if (i2c_device[0x0F] == 0) {
    Serial.println("No calibration");
  }
  if (i2c_device[0x0F] == 3 ) {
    Serial.println("Single point calibration");
  }
  if (i2c_device[0x0F] == 13 ) {
    Serial.println("High value of a two point calibration");
  }

  Serial.print("Temperature Compensation: ");
  temp_comp = read_four_bytes(0x10);
  Serial.println((float)temp_comp / 100);

  Serial.println("");
  Serial.println("");
}

void device_commands() {

  Serial.println("i read device info.");
  Serial.println("This will read and display the status of the device (registers 0x00- 0x13)");
  Serial.println("");
  Serial.println("k,XXX set the probe type. Where XXX is the K constant of the EC probe");
  Serial.println("k,5<CR> to set the probe type to K=5.0");
  Serial.println("");
  Serial.println("t,XXX set the temperature compensation. Where XXX is the temperature in C");
  Serial.println("t,35.8<CR> to set the temperature compensation to 35.8 C");
  Serial.println("");
  Serial.println("calval,XXX set the value the probe will be calibrated to");
  Serial.println("calval,12880<CR> to set the intended calibration value to 12880us");
  Serial.println("");
  Serial.println("cal,clear<CR> to clear all previous calibrations");
  Serial.println("cal,dry<CR> to do a dry calibration");
  Serial.println("cal,one<CR> to do a single point calibration");
  Serial.println("cal,low<CR> to do the first part of a 2 part calibration, the low point calibration");
  Serial.println("cal,high<CR> to do the second part of a 2 part calibration, the high point calibration");
  Serial.println("");
  Serial.println("on this will enable readings and set the LED to change its state per reading");
  Serial.println("on<CR> the device is now on and actively taking readings");
  Serial.println("");
  Serial.println("off this will disable readings and turn off the LED");
  Serial.println("off<CR> the device is now off");
  Serial.println("");
  Serial.println("r take a reading");
  Serial.println("r<CR>");
}

//this interrupt will trigger when the data coming from the serial monitor is received
//we read the data sent from the serial monitor until we see a <CR>. We also count how many characters have been received
//this stops the buffer from transmitting leftovers or garbage.
//this is used as a flag to indicate that we have received a string

//this function reads registers from the OEM EC "i2c_read( register to start from, number  of bytes to read)"
//counter
//call the device by its ID number
//transmit the register that we will start from
//end the I2C data transmission

//call the device and request to read some bytes
//this loop will read OEM register bytes starting at X register and ending at the number of bytes requested
//read a byte, load it into an array

//end the I2C data transmission

//we make an 20 byte character array to hold incoming data from the serial monitor
//char pointer used in string parsing
//char pointer used in string parsing
//we need to know how many characters have been received.
//a flag to signal when data has been received from the serial monitor
//a counter so we know how many bytes were received from the serial monitor
//the array that holds the register data from the EC OEM

After flashing the Arduino with this code, open the serial monitor. 
Set the serial monitor to append carriage return only and set the 
baud rate to 9600. Send a "?" to see a list of commands.  

https://www.atlas-scientific.com/_files/code/ino_files/OEM_EC_sample_code.zip

